Rules for integrands of the form $u\,\text{Trig}[d\,(a+b\,\text{Log}[c\,x^n])]^p$

1. $\int u\,\text{Sin}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$

   1. $\int \text{Sin}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$

      1. $\int \text{Sin}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$ when $p\in\mathbb{Z}^+\land b^2 d^2 n^2 p^2+1\neq 0$

         1: $\int \text{Sin}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx$ when $b^2 d^2 n^2+1\neq 0$

Rule: If $b^2 d^2 n^2+1\neq 0$, then

$$\int \text{Sin}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx\longrightarrow\frac{x\,\text{Sin}[d\,(a+b\,\text{Log}[c\,x^n])]}{b^2 d^2 n^2+1}-\frac{b\,d\,n\,x\,\text{Cos}[d\,(a+b\,\text{Log}[c\,x^n])]}{b^2 d^2 n^2+1}$$

Program code:

```
Int[Sin[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  x*Sin[d*(a+b*Log[c*x^n])]/(b^2*d^2*n^2+1) -
  b*d*n*x*Cos[d*(a+b*Log[c*x^n])]/(b^2*d^2*n^2+1) /;
FreeQ[{a,b,c,d,n},x] && NeQ[b^2*d^2*n^2+1,0]
```

```
Int[Cos[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  x*Cos[d*(a+b*Log[c*x^n])]/(b^2*d^2*n^2+1) +
  b*d*n*x*Sin[d*(a+b*Log[c*x^n])]/(b^2*d^2*n^2+1) /;
FreeQ[{a,b,c,d,n},x] && NeQ[b^2*d^2*n^2+1,0]
```

2: $\int \sin\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p dx$ when $p-1\in\mathbb{Z}^+ \wedge b^2 d^2 n^2 p^2+1\neq 0$

Rule: If $p-1\in\mathbb{Z}^+ \wedge b^2 d^2 n^2 p^2+1\neq 0$, then

$$\int \sin\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p dx \longrightarrow$$

$$\frac{x\sin\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p}{b^2 d^2 n^2 p^2+1}-\frac{b\,d\,n\,p\,x\cos\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]\sin\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^{p-1}}{b^2 d^2 n^2 p^2+1}+\frac{b^2 d^2 n^2 p\,(p-1)}{b^2 d^2 n^2 p^2+1}\int \sin\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^{p-2} dx$$

Program code:

```
Int[Sin[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_,x_Symbol] :=
  x*Sin[d*(a+b*Log[c*x^n])]^p/(b^2*d^2*n^2*p^2+1) -
  b*d*n*p*x*Cos[d*(a+b*Log[c*x^n])]*Sin[d*(a+b*Log[c*x^n])]^(p-1)/(b^2*d^2*n^2*p^2+1) +
  b^2*d^2*n^2*p*(p-1)/(b^2*d^2*n^2*p^2+1)*Int[Sin[d*(a+b*Log[c*x^n])]^(p-2),x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[p,1] && NeQ[b^2*d^2*n^2*p^2+1,0]
```

```
Int[Cos[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_,x_Symbol] :=
  x*Cos[d*(a+b*Log[c*x^n])]^p/(b^2*d^2*n^2*p^2+1) +
  b*d*n*p*x*Cos[d*(a+b*Log[c*x^n])]^(p-1)*Sin[d*(a+b*Log[c*x^n])]/(b^2*d^2*n^2*p^2+1) +
  b^2*d^2*n^2*p*(p-1)/(b^2*d^2*n^2*p^2+1)*Int[Cos[d*(a+b*Log[c*x^n])]^(p-2),x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[p,1] && NeQ[b^2*d^2*n^2*p^2+1,0]
```

2. $\int \text{Sin}[d\ (a+b\,\text{Log}[x])]^p\, dx$

1: $\int \text{Sin}[d\ (a+b\,\text{Log}[x])]^p\, dx$ when $p\in\mathbb{Z}^+ \wedge b^2 d^2 p^2+1==0$

- Derivation: Algebraic expansion

- Basis: If $b^2 d^2 p^2+1==0 \wedge p\in\mathbb{Z}$, then $\text{sin}[d\ (a+b\,\text{Log}[x])]^p == \frac{1}{2^p b^p d^p p^p}\left(e^{a b d^2 p}\, x^{-\frac{1}{p}} - e^{-a b d^2 p}\, x^{\frac{1}{p}}\right)^p$

- Basis: If $b^2 d^2 p^2+1==0 \wedge p\in\mathbb{Z}$, then $\text{cos}[d\ (a+b\,\text{Log}[x])]^p == \frac{1}{2^p}\left(e^{a b d^2 p}\, x^{-\frac{1}{p}} + e^{-a b d^2 p}\, x^{\frac{1}{p}}\right)^p$

- Note: The above identities need to be formally derived, and possibly the domain of p expanded.

- Rule: If $p\in\mathbb{Z}^+ \wedge b^2 d^2 p^2+1==0$, then

$$\int \text{Sin}[d\ (a+b\,\text{Log}[x])]^p\, dx \longrightarrow \frac{1}{2^p b^p d^p p^p}\int \text{ExpandIntegrand}\left[\left(e^{a b d^2 p}\, x^{-\frac{1}{p}} - e^{-a b d^2 p}\, x^{\frac{1}{p}}\right)^p,\ x\right] dx$$

Program code:

```
Int[Sin[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  1/(2^p*b^p*d^p*p^p)*Int[ExpandIntegrand[(E^(a*b*d^2*p)*x^(-1/p)-E^(-a*b*d^2*p)*x^(1/p))^p,x],x] /;
FreeQ[{a,b,d},x] && IGtQ[p,0] && EqQ[b^2*d^2*p^2+1,0]
```

```
Int[Cos[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  1/2^p*Int[ExpandIntegrand[(E^(a*b*d^2*p)*x^(-1/p)+E^(-a*b*d^2*p)*x^(1/p))^p,x],x] /;
FreeQ[{a,b,d},x] && IGtQ[p,0] && EqQ[b^2*d^2*p^2+1,0]
```

x: $\int \mathrm{Sin}[d\,(a+b\,\mathrm{Log}[x])]^p\,dx$ when $p \in \mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\mathrm{Sin}[d\,(a+b\,\mathrm{Log}[x])] == \frac{1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}}{-2\,i\,e^{i\,a\,d}\,x^{i\,b\,d}}$

Basis: $\mathrm{Cos}[d\,(a+b\,\mathrm{Log}[x])] == \frac{1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}}{2\,e^{i\,a\,d}\,x^{i\,b\,d}}$

Rule: If $p \in \mathbb{Z}$, then

$$\int \mathrm{Sin}[d\,(a+b\,\mathrm{Log}[x])]^p\,dx \to \frac{1}{(-2\,i)^p\,e^{i\,a\,d\,p}}\int \frac{\left(1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}{x^{i\,b\,d\,p}}\,dx$$

Program code:

```
(* Int[Sin[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  1/((-2*I)^p*E^(I*a*d*p))*Int[(1-E^(2*I*a*d)*x^(2*I*b*d))^p/x^(I*b*d*p),x] /;
FreeQ[{a,b,d},x] && IntegerQ[p] *)
```

```
(* Int[Cos[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  1/(2^p*E^(I*a*d*p))*Int[(1+E^(2*I*a*d)*x^(2*I*b*d))^p/x^(I*b*d*p),x] /;
FreeQ[{a,b,d},x] && IntegerQ[p] *)
```

2: $\int \mathrm{Sin}[d\,(a+b\,\mathrm{Log}[x])]^p\,dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\mathrm{Sin}[d\,(a+b\,\mathrm{Log}[x])]^p\,x^{i\,b\,d\,p}}{\left(1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p} == 0$

Basis: $\partial_x \frac{\mathrm{Cos}[d\,(a+b\,\mathrm{Log}[x])]^p\,x^{i\,b\,d\,p}}{\left(1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p} == 0$

Rule: If $p \notin \mathbb{Z}$, then

$$\int \mathrm{Sin}[d\,(a+b\,\mathrm{Log}[x])]^p\,dx \longrightarrow \frac{\mathrm{Sin}[d\,(a+b\,\mathrm{Log}[x])]^p\,x^{i\,b\,d\,p}}{\left(1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}\int \frac{\left(1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}{x^{i\,b\,d\,p}}\,dx$$

Program code:

```
Int[Sin[d_.*(a_.+b_.*Log[x_])]^p_,x_Symbol] :=
  Sin[d*(a+b*Log[x])]^p*x^(I*b*d*p)/(1-E^(2*I*a*d)*x^(2*I*b*d))^p*
    Int[(1-E^(2*I*a*d)*x^(2*I*b*d))^p/x^(I*b*d*p),x] /;
FreeQ[{a,b,d,p},x] && Not[IntegerQ[p]]
```

```
Int[Cos[d_.*(a_.+b_.*Log[x_])]^p_,x_Symbol] :=
  Cos[d*(a+b*Log[x])]^p*x^(I*b*d*p)/(1+E^(2*I*a*d)*x^(2*I*b*d))^p*
    Int[(1+E^(2*I*a*d)*x^(2*I*b*d))^p/x^(I*b*d*p),x] /;
FreeQ[{a,b,d,p},x] && Not[IntegerQ[p]]
```

3: $\int \mathrm{Sin}\left[d\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)\right]^p\,dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{x}{\left(c\,x^n\right)^{1/n}} == 0$

Basis: $\frac{F[c\,x^n]}{x} == \frac{1}{n}\,\mathrm{Subst}\left[\frac{F[x]}{x},\,x,\,c\,x^n\right]\partial_x\left(c\,x^n\right)$

Rule:

$$\int \mathrm{Sin}\left[d\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)\right]^p\,dx \longrightarrow \frac{x}{\left(c\,x^n\right)^{1/n}}\int \frac{\left(c\,x^n\right)^{1/n}\,\mathrm{Sin}\left[d\left(a+b\,\mathrm{Log}\left[c\,x^n\right]\right)\right]^p}{x}\,dx$$

$$\longrightarrow \frac{x}{n\left(c\,x^n\right)^{1/n}}\,\mathrm{Subst}\left[\int x^{1/n-1}\,\mathrm{Sin}[d\,(a+b\,\mathrm{Log}[x])]^p\,dx,\,x,\,c\,x^n\right]$$

```
Int[Sin[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  x/(n*(c*x^n)^(1/n))*Subst[Int[x^(1/n-1)*Sin[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

```
Int[Cos[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  x/(n*(c*x^n)^(1/n))*Subst[Int[x^(1/n-1)*Cos[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

2. $\int (e\,x)^m \operatorname{Sin}\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p dx$

1. $\int (e\,x)^m \operatorname{Sin}\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p dx$ when $p\in\mathbb{Z}^+\wedge b^2 d^2 n^2 p^2+(m+1)^2\neq 0$

1: $\int (e\,x)^m \operatorname{Sin}\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right] dx$ when $b^2 d^2 n^2+(m+1)^2\neq 0$

Rule: If $b^2 d^2 n^2+(m+1)^2\neq 0$, then

$$\int (e\,x)^m \operatorname{Sin}\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right] dx \longrightarrow \frac{(m+1)\ (e\,x)^{m+1}\operatorname{Sin}\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]}{b^2 d^2 e\,n^2+e\ (m+1)^2}-\frac{b\,d\,n\ (e\,x)^{m+1}\operatorname{Cos}\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]}{b^2 d^2 e\,n^2+e\ (m+1)^2}$$

Program code:

```
Int[(e_.*x_)^m_.*Sin[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (m+1)*(e*x)^(m+1)*Sin[d*(a+b*Log[c*x^n])]/(b^2*d^2*e*n^2+e*(m+1)^2) -
  b*d*n*(e*x)^(m+1)*Cos[d*(a+b*Log[c*x^n])]/(b^2*d^2*e*n^2+e*(m+1)^2) /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[b^2*d^2*n^2+(m+1)^2,0]
```

```
Int[(e_.*x_)^m_.*Cos[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  (m+1)*(e*x)^(m+1)*Cos[d*(a+b*Log[c*x^n])]/(b^2*d^2*e*n^2+e*(m+1)^2) +
  b*d*n*(e*x)^(m+1)*Sin[d*(a+b*Log[c*x^n])]/(b^2*d^2*e*n^2+e*(m+1)^2) /;
FreeQ[{a,b,c,d,e,m,n},x] && NeQ[b^2*d^2*n^2+(m+1)^2,0]
```

2: $\int (e\,x)^m \operatorname{Sin}\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p dx$ when $p-1\in\mathbb{Z}^+\wedge b^2 d^2 n^2 p^2+(m+1)^2\neq 0$

Rule: If $p-1\in\mathbb{Z}^+\wedge b^2 d^2 n^2 p^2+(m+1)^2\neq 0$, then

$$\int (e\,x)^m \operatorname{Sin}\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p dx \longrightarrow$$

$$\frac{(m+1)\ (e\,x)^{m+1}\,\text{Sin}\left[d\ \left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p}{b^2\,d^2\,e\,n^2\,p^2+e\ (m+1)^2}-\frac{b\,d\,n\,p\ (e\,x)^{m+1}\,\text{Cos}\left[d\ \left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]\,\text{Sin}\left[d\ \left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^{p-1}}{b^2\,d^2\,e\,n^2\,p^2+e\ (m+1)^2}+$$
$$\frac{b^2\,d^2\,n^2\,p\ (p-1)}{b^2\,d^2\,n^2\,p^2+\ (m+1)^2}\int (e\,x)^m\,\text{Sin}\left[d\ \left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^{p-2}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sin[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_,x_Symbol] :=
  (m+1)*(e*x)^(m+1)*Sin[d*(a+b*Log[c*x^n])]^p/(b^2*d^2*e*n^2*p^2+e*(m+1)^2) -
  b*d*n*p*(e*x)^(m+1)*Cos[d*(a+b*Log[c*x^n])]*Sin[d*(a+b*Log[c*x^n])]^(p-1)/(b^2*d^2*e*n^2*p^2+e*(m+1)^2) +
  b^2*d^2*n^2*p*(p-1)/(b^2*d^2*n^2*p^2+(m+1)^2)*Int[(e*x)^m*Sin[d*(a+b*Log[c*x^n])]^(p-2),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IGtQ[p,1] && NeQ[b^2*d^2*n^2*p^2+(m+1)^2,0]
```

```
Int[(e_.*x_)^m_.*Cos[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_,x_Symbol] :=
  (m+1)*(e*x)^(m+1)*Cos[d*(a+b*Log[c*x^n])]^p/(b^2*d^2*e*n^2*p^2+e*(m+1)^2) +
  b*d*n*p*(e*x)^(m+1)*Sin[d*(a+b*Log[c*x^n])]*Cos[d*(a+b*Log[c*x^n])]^(p-1)/(b^2*d^2*e*n^2*p^2+e*(m+1)^2) +
  b^2*d^2*n^2*p*(p-1)/(b^2*d^2*n^2*p^2+(m+1)^2)*Int[(e*x)^m*Cos[d*(a+b*Log[c*x^n])]^(p-2),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && IGtQ[p,1] && NeQ[b^2*d^2*n^2*p^2+(m+1)^2,0]
```

2. $\int (e\,x)^m \,\mathrm{Sin}[d\,(a+b\,\mathrm{Log}[x])]^p\,dx$

1: $\int (e\,x)^m \,\mathrm{Sin}[d\,(a+b\,\mathrm{Log}[x])]^p\,dx$ when $p\in\mathbb{Z}^+ \wedge b^2 d^2 p^2+(m+1)^2==0$

Derivation: Algebraic expansion

Basis: If $b^2 d^2 p^2+(m+1)^2==0 \wedge p\in\mathbb{Z}$, then $\mathrm{Sin}[d\,(a+b\,\mathrm{Log}[x])]^p == \frac{(m+1)^p}{2^p b^p d^p p^p}\left(e^{\frac{a b d^2 p}{m+1}} x^{-\frac{m+1}{p}} - e^{-\frac{a b d^2 p}{m+1}} x^{\frac{m+1}{p}}\right)^p$

Basis: If $b^2 d^2 p^2+(m+1)^2==0 \wedge p\in\mathbb{Z}$, then $\mathrm{Cos}[d\,(a+b\,\mathrm{Log}[x])]^p == \frac{1}{2^p}\left(e^{\frac{a b d^2 p}{m+1}} x^{-\frac{m+1}{p}} + e^{-\frac{a b d^2 p}{m+1}} x^{\frac{m+1}{p}}\right)^p$

Note: The above identities need to be formally derived, and possibly the domain of p expanded.

Rule: If $p\in\mathbb{Z}^+ \wedge b^2 d^2 p^2+(m+1)^2==0$, then

$$\int (e\,x)^m \,\mathrm{Sin}[d\,(a+b\,\mathrm{Log}[x])]^p\,dx \to \frac{(m+1)^p}{2^p b^p d^p p^p}\int \mathrm{ExpandIntegrand}\left[(e\,x)^m \left(e^{\frac{a b d^2 p}{m+1}} x^{-\frac{m+1}{p}} - e^{-\frac{a b d^2 p}{m+1}} x^{\frac{m+1}{p}}\right)^p,\,x\right]dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sin[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  (m+1)^p/(2^p*b^p*d^p*p^p)*
    Int[ExpandIntegrand[(e*x)^m*(E^(a*b*d^2*p/(m+1))*x^(-(m+1)/p)-E^(-a*b*d^2*p/(m+1))*x^((m+1)/p))^p,x],x] /;
FreeQ[{a,b,d,e,m},x] && IGtQ[p,0] && EqQ[b^2*d^2*p^2+(m+1)^2,0]
```

```
Int[(e_.*x_)^m_.*Cos[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  1/2^p*Int[ExpandIntegrand[(e*x)^m*(E^(a*b*d^2*p/(m+1))*x^(-(m+1)/p)+E^(-a*b*d^2*p/(m+1))*x^((m+1)/p))^p,x],x] /;
FreeQ[{a,b,d,e,m},x] && IGtQ[p,0] && EqQ[b^2*d^2*p^2+(m+1)^2,0]
```

x: $\int (e\,x)^m \operatorname{Sin}[d\,(a+b\operatorname{Log}[x])]^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\operatorname{Sin}\left[d\,\left(a+b\operatorname{Log}[x]\right)\right] = \frac{1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}}{-2\,i\,e^{i\,a\,d}\,x^{i\,b\,d}}$

Basis: $\operatorname{Cos}\left[d\,\left(a+b\operatorname{Log}[x]\right)\right] = \frac{1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}}{2\,e^{i\,a\,d}\,x^{i\,b\,d}}$

Rule: If $p\in\mathbb{Z}$, then

$$\int (e\,x)^m \operatorname{Sin}[d\,(a+b\operatorname{Log}[x])]^p\,dx \longrightarrow \frac{1}{(-2\,i)^p\,e^{i\,a\,d\,p}}\int \frac{(e\,x)^m\left(1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}{x^{i\,b\,d\,p}}\,dx$$

Program code:

```
(* Int[(e_.*x_)^m_.*Sin[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  1/((-2*I)^p*E^(I*a*d*p))*Int[(e*x)^m*(1-E^(2*I*a*d)*x^(2*I*b*d))^p/x^(I*b*d*p),x] /;
FreeQ[{a,b,d,e,m},x] && IntegerQ[p] *)
```

```
(* Int[(e_.*x_)^m_.*Cos[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  1/(2^p*E^(I*a*d*p))*Int[(e*x)^m*(1+E^(2*I*a*d)*x^(2*I*b*d))^p/x^(I*b*d*p),x] /;
FreeQ[{a,b,d,e,m},x] && IntegerQ[p] *)
```

2: $\int (e\,x)^m \operatorname{Sin}[d\,(a+b\operatorname{Log}[x])]^p\,dx$ when $p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\operatorname{Sin}[d\,(a+b\operatorname{Log}[x])]^p\,x^{i\,b\,d\,p}}{\left(1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p} = 0$

Basis: $\partial_x \frac{\operatorname{Cos}[d\,(a+b\operatorname{Log}[x])]^p\,x^{i\,b\,d\,p}}{\left(1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p} = 0$

Rule: If $p\notin\mathbb{Z}$, then

$$\int (e\,x)^m \,\text{Sin}[d\,(a+b\,\text{Log}[x])]^p \,dx \longrightarrow \frac{\text{Sin}[d\,(a+b\,\text{Log}[x])]^p\,x^{i\,b\,d\,p}}{\left(1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}\int \frac{(e\,x)^m\left(1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}{x^{i\,b\,d\,p}}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sin[d_.*(a_.+b_.*Log[x_])]^p_,x_Symbol] :=
  Sin[d*(a+b*Log[x])]^p*x^(I*b*d*p)/(1-E^(2*I*a*d)*x^(2*I*b*d))^p*
    Int[(e*x)^m*(1-E^(2*I*a*d)*x^(2*I*b*d))^p/x^(I*b*d*p),x] /;
FreeQ[{a,b,d,e,m,p},x] && Not[IntegerQ[p]]
```

```
Int[(e_.*x_)^m_.*Cos[d_.*(a_.+b_.*Log[x_])]^p_,x_Symbol] :=
  Cos[d*(a+b*Log[x])]^p*x^(I*b*d*p)/(1+E^(2*I*a*d)*x^(2*I*b*d))^p*
    Int[(e*x)^m*(1+E^(2*I*a*d)*x^(2*I*b*d))^p/x^(I*b*d*p),x] /;
FreeQ[{a,b,d,e,m,p},x] && Not[IntegerQ[p]]
```

3: $\int (e\,x)^m \,\text{Sin}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p \,dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{x}{(c\,x^n)^{1/n}} == 0$

Basis: $\frac{F[c\,x^n]}{x} == \frac{1}{n}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,c\,x^n\right]\,\partial_x\left(c\,x^n\right)$

Rule:

$$\int (e\,x)^m \,\text{Sin}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p \,dx \longrightarrow \frac{(e\,x)^{m+1}}{e\,\left(c\,x^n\right)^{(m+1)/n}}\int \frac{\left(c\,x^n\right)^{(m+1)/n}\,\text{Sin}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p}{x}\,dx$$

$$\longrightarrow \frac{(e\,x)^{m+1}}{e\,n\,\left(c\,x^n\right)^{(m+1)/n}}\,\text{Subst}\left[\int x^{(m+1)/n-1}\,\text{Sin}[d\,(a+b\,\text{Log}[x])]^p \,dx,\,x,\,c\,x^n\right]$$

```
Int[(e_.*x_)^m_.*Sin[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  (e*x)^(m+1)/(e*n*(c*x^n)^((m+1)/n))*Subst[Int[x^((m+1)/n-1)*Sin[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

```
Int[(e_.*x_)^m_.*Cos[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  (e*x)^(m+1)/(e*n*(c*x^n)^((m+1)/n))*Subst[Int[x^((m+1)/n-1)*Cos[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

3: $\int (h\,(e+f\,\text{Log}[g\,x^m]))^q\,\text{Sin}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $\text{Sin}[d\,(a+b\,\text{Log}[z])] = \frac{i}{2}\,e^{-i\,a\,d}\,z^{-i\,b\,d} - \frac{i}{2}\,e^{i\,a\,d}\,z^{i\,b\,d}$

Basis: $\text{Cos}[d\,(a+b\,\text{Log}[z])] = \frac{1}{2}\,e^{-i\,a\,d}\,z^{-i\,b\,d} + \frac{1}{2}\,e^{i\,a\,d}\,z^{i\,b\,d}$

Rule:

$$\int (h\,(e+f\,\text{Log}[g\,x^m]))^q\,\text{Sin}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx \longrightarrow$$
$$\frac{i\,e^{-i\,a\,d}\,(c\,x^n)^{-i\,b\,d}}{2\,x^{-i\,b\,d\,n}}\int x^{-i\,b\,d\,n}\,(h\,(e+f\,\text{Log}[g\,x^m]))^q\,dx - \frac{i\,e^{i\,a\,d}\,(c\,x^n)^{i\,b\,d}}{2\,x^{i\,b\,d\,n}}\int x^{i\,b\,d\,n}\,(h\,(e+f\,\text{Log}[g\,x^m]))^q\,dx$$

Program code:

```
Int[(h_.*(e_.+f_.*Log[g_.*x_^m_.]))^q_.*Sin[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  I*E^(-I*a*d)*(c*x^n)^(-I*b*d)/(2*x^(-I*b*d*n))*Int[x^(-I*b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] -
  I*E^(I*a*d)*(c*x^n)^(I*b*d)/(2*x^(I*b*d*n))*Int[x^(I*b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,q},x]
```

```
Int[(h_.*(e_.+f_.*Log[g_.*x_^m_.]))^q_.*Cos[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  E^(-I*a*d)*(c*x^n)^(-I*b*d)/(2*x^(-I*b*d*n))*Int[x^(-I*b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] +
  E^(I*a*d)*(c*x^n)^(I*b*d)/(2*x^(I*b*d*n))*Int[x^(I*b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,q},x]
```

4: $\int (i\,x)^r\,(h\,(e+f\,\text{Log}[g\,x^m]))^q\,\text{Sin}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $\text{Sin}[d\,(a+b\,\text{Log}[z])] == \frac{i}{2}\,e^{-i\,a\,d}\,z^{-i\,b\,d} - \frac{i}{2}\,e^{i\,a\,d}\,z^{i\,b\,d}$

Basis: $\text{Cos}[d\,(a+b\,\text{Log}[z])] == \frac{1}{2}\,e^{-i\,a\,d}\,z^{-i\,b\,d} + \frac{1}{2}\,e^{i\,a\,d}\,z^{i\,b\,d}$

Rule:

$$\int (i\,x)^r\,(h\,(e+f\,\text{Log}[g\,x^m]))^q\,\text{Sin}[d\,(a+b\,\text{Log}[c\,x^n])]\,dx \longrightarrow$$

$$\frac{i\,e^{-i\,a\,d}\,(i\,x)^r\,(c\,x^n)^{-i\,b\,d}}{2\,x^{r-i\,b\,d\,n}}\int x^{r-i\,b\,d\,n}\,(h\,(e+f\,\text{Log}[g\,x^m]))^q\,dx - \frac{i\,e^{i\,a\,d}\,(i\,x)^r\,(c\,x^n)^{i\,b\,d}}{2\,x^{r+i\,b\,d\,n}}\int x^{r+i\,b\,d\,n}\,(h\,(e+f\,\text{Log}[g\,x^m]))^q\,dx$$

Program code:

```
Int[(i_.*x_)^r_.*(h_.*(e_.+f_.*Log[g_.*x_^m_.]))^q_.*Sin[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  I*E^(-I*a*d)*(i*x)^r*(c*x^n)^(-I*b*d)/(2*x^(r-I*b*d*n))*Int[x^(r-I*b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] -
  I*E^(I*a*d)*(i*x)^r*(c*x^n)^(I*b*d)/(2*x^(r+I*b*d*n))*Int[x^(r+I*b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,m,n,q,r},x]
```

```
Int[(i_.*x_)^r_.*(h_.*(e_.+f_.*Log[g_.*x_^m_.]))^q_.*Cos[d_.*(a_.+b_.*Log[c_.*x_^n_.])],x_Symbol] :=
  E^(-I*a*d)*(i*x)^r*(c*x^n)^(-I*b*d)/(2*x^(r-I*b*d*n))*Int[x^(r-I*b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] +
  E^(I*a*d)*(i*x)^r*(c*x^n)^(I*b*d)/(2*x^(r+I*b*d*n))*Int[x^(r+I*b*d*n)*(h*(e+f*Log[g*x^m]))^q,x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,m,n,q,r},x]
```

2. $\int u \operatorname{Tan}[d (a + b \operatorname{Log}[c x^n])]^p \, dx$

1. $\int \operatorname{Tan}[d (a + b \operatorname{Log}[c x^n])]^p \, dx$

1: $\int \operatorname{Tan}[d (a + b \operatorname{Log}[x])]^p \, dx$

–

Derivation: Algebraic expansion

Basis: $\operatorname{Tan}[z] == \frac{i - i e^{2 i z}}{1 + e^{2 i z}}$

–

Basis: $\operatorname{Cot}[z] == \frac{-i - i e^{2 i z}}{1 - e^{2 i z}}$

–

Rule:

$$\int \operatorname{Tan}[d (a + b \operatorname{Log}[x])]^p \, dx \longrightarrow \int \left(\frac{i - i e^{2 i a d} x^{2 i b d}}{1 + e^{2 i a d} x^{2 i b d}}\right)^p dx$$

Program code:

```
Int[Tan[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  Int[((I-I*E^(2*I*a*d)*x^(2*I*b*d))/(1+E^(2*I*a*d)*x^(2*I*b*d)))^p,x] /;
FreeQ[{a,b,d,p},x]
```

```
Int[Cot[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  Int[((-I-I*E^(2*I*a*d)*x^(2*I*b*d))/(1-E^(2*I*a*d)*x^(2*I*b*d)))^p,x] /;
FreeQ[{a,b,d,p},x]
```

2: $\int \text{Tan}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{x}{(c\,x^n)^{1/n}} == 0$

Basis: $\frac{F[c\,x^n]}{x} == \frac{1}{n}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,c\,x^n\right]\,\partial_x\,(c\,x^n)$

Rule:

$$\int \text{Tan}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx \to \frac{x}{(c\,x^n)^{1/n}}\int \frac{(c\,x^n)^{1/n}\,\text{Tan}[d\,(a+b\,\text{Log}[c\,x^n])]^p}{x}\,dx$$

$$\to \frac{x}{n\,(c\,x^n)^{1/n}}\,\text{Subst}\left[\int x^{1/n-1}\,\text{Tan}[d\,(a+b\,\text{Log}[x])]^p\,dx,\,x,\,c\,x^n\right]$$

```
Int[Tan[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  x/(n*(c*x^n)^(1/n))*Subst[Int[x^(1/n-1)*Tan[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

```
Int[Cot[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  x/(n*(c*x^n)^(1/n))*Subst[Int[x^(1/n-1)*Cot[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

2. $\int (e\,x)^m \operatorname{Tan}\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p\,dx$

1: $\int (e\,x)^m \operatorname{Tan}[d\,(a+b\operatorname{Log}[x])]^p\,dx$

Derivation: Algebraic expansion

Basis: $\operatorname{Tan}[z] == \frac{i-i\,e^{2\,i\,z}}{1+e^{2\,i\,z}}$

Basis: $\operatorname{Cot}[z] == \frac{-i-i\,e^{2\,i\,z}}{1-e^{2\,i\,z}}$

Rule:

$$\int (e\,x)^m \operatorname{Tan}[d\,(a+b\operatorname{Log}[x])]^p\,dx \longrightarrow \int (e\,x)^m \left(\frac{i-i\,e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}}{1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}}\right)^p dx$$

Program code:

```
Int[(e_.*x_)^m_.*Tan[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  Int[(e*x)^m*((I-I*E^(2*I*a*d)*x^(2*I*b*d))/(1+E^(2*I*a*d)*x^(2*I*b*d)))^p,x] /;
FreeQ[{a,b,d,e,m,p},x]
```

```
Int[(e_.*x_)^m_.*Cot[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  Int[(e*x)^m*((-I-I*E^(2*I*a*d)*x^(2*I*b*d))/(1-E^(2*I*a*d)*x^(2*I*b*d)))^p,x] /;
FreeQ[{a,b,d,e,m,p},x]
```

2: $\int (e\,x)^m \operatorname{Tan}\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p dx$

- Derivation: Piecewise constant extraction and integration by substitution

- Basis: $\partial_x \frac{x}{\left(c\,x^n\right)^{1/n}} = 0$

- Basis: $\frac{F\left[c\,x^n\right]}{x} = \frac{1}{n}\operatorname{Subst}\left[\frac{F[x]}{x}, x, c\,x^n\right] \partial_x\left(c\,x^n\right)$

- Rule:

$$\int (e\,x)^m \operatorname{Tan}\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p dx \longrightarrow \frac{(e\,x)^{m+1}}{e\left(c\,x^n\right)^{(m+1)/n}} \int \frac{\left(c\,x^n\right)^{(m+1)/n} \operatorname{Tan}\left[d\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p}{x} dx$$

$$\longrightarrow \frac{(e\,x)^{m+1}}{e\,n\left(c\,x^n\right)^{(m+1)/n}} \operatorname{Subst}\left[\int x^{(m+1)/n-1} \operatorname{Tan}\left[d\left(a+b\operatorname{Log}[x]\right)\right]^p dx, x, c\,x^n\right]$$

```
Int[(e_.*x_)^m_.*Tan[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  (e*x)^(m+1)/(e*n*(c*x^n)^((m+1)/n))*Subst[Int[x^((m+1)/n-1)*Tan[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

```
Int[(e_.*x_)^m_.*Cot[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  (e*x)^(m+1)/(e*n*(c*x^n)^((m+1)/n))*Subst[Int[x^((m+1)/n-1)*Cot[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

3. $\int u\,\text{Sec}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p\,dx$

1. $\int \text{Sec}\left[d\,\left(a+b\,\text{Log}\left[c\,x^n\right]\right)\right]^p\,dx$

1. $\int \text{Sec}[d\,(a+b\,\text{Log}[x])]^p\,dx$

1: $\int \text{Sec}[d\,(a+b\,\text{Log}[x])]^p\,dx$ when $p\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\text{Sec}\left[d\,\left(a+b\,\text{Log}[x]\right)\right]==\frac{2\,e^{i\,a\,d}\,x^{i\,b\,d}}{1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}}$

Basis: $\text{Csc}\left[d\,\left(a+b\,\text{Log}[x]\right)\right]==-\frac{2\,i\,e^{i\,a\,d}\,x^{i\,b\,d}}{1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}}$

Rule: If $p\in\mathbb{Z}$, then

$$\int \text{Sec}[d\,(a+b\,\text{Log}[x])]^p\,dx\;\longrightarrow\;2^p\,e^{i\,a\,d\,p}\int \frac{x^{i\,b\,d\,p}}{\left(1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}\,dx$$

Program code:

```
Int[Sec[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  2^p*E^(I*a*d*p)*Int[x^(I*b*d*p)/(1+E^(2*I*a*d)*x^(2*I*b*d))^p,x] /;
FreeQ[{a,b,d},x] && IntegerQ[p]
```

```
Int[Csc[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  (-2*I)^p*E^(I*a*d*p)*Int[x^(I*b*d*p)/(1-E^(2*I*a*d)*x^(2*I*b*d))^p,x] /;
FreeQ[{a,b,d},x] && IntegerQ[p]
```

2: $\int \text{Sec}[d\,(a+b\,\text{Log}[x])]^p\,dx$ when $p \notin \mathbb{Z}$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $\partial_x \frac{\text{Sec}[d\,(a+b\,\text{Log}[x])]^p\,\left(1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}{x^{i\,b\,d\,p}} == 0$

Basis: $\partial_x \frac{\text{Csc}[d\,(a+b\,\text{Log}[x])]^p\,\left(1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}{x^{i\,b\,d\,p}} == 0$

Rule: If $p \notin \mathbb{Z}$, then

$$\int \text{Sec}[d\,(a+b\,\text{Log}[x])]^p\,dx \longrightarrow \frac{\text{Sec}[d\,(a+b\,\text{Log}[x])]^p\,\left(1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}{x^{i\,b\,d}} \int \frac{x^{i\,b\,d\,p}}{\left(1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}\,dx$$

Program code:

```
Int[Sec[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  Sec[d*(a+b*Log[x])]^p*(1+E^(2*I*a*d)*x^(2*I*b*d))^p/x^(I*b*d*p)*
    Int[x^(I*b*d*p)/(1+E^(2*I*a*d)*x^(2*I*b*d))^p,x] /;
FreeQ[{a,b,d,p},x] && Not[IntegerQ[p]]
```

```
Int[Csc[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  Csc[d*(a+b*Log[x])]^p*(1-E^(2*I*a*d)*x^(2*I*b*d))^p/x^(I*b*d*p)*
    Int[x^(I*b*d*p)/(1-E^(2*I*a*d)*x^(2*I*b*d))^p,x] /;
FreeQ[{a,b,d,p},x] && Not[IntegerQ[p]]
```

2: $\int \text{Sec}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{x}{(c\,x^n)^{1/n}} == 0$

Basis: $\frac{F[c\,x^n]}{x} == \frac{1}{n}\,\text{Subst}\left[\frac{F[x]}{x},\,x,\,c\,x^n\right]\,\partial_x\,(c\,x^n)$

Rule:

$$\int \text{Sec}[d\,(a+b\,\text{Log}[c\,x^n])]^p\,dx \to \frac{x}{(c\,x^n)^{1/n}}\int \frac{(c\,x^n)^{1/n}\,\text{Sec}[d\,(a+b\,\text{Log}[c\,x^n])]^p}{x}\,dx$$

$$\to \frac{x}{n\,(c\,x^n)^{1/n}}\,\text{Subst}\left[\int x^{1/n-1}\,\text{Sec}[d\,(a+b\,\text{Log}[x])]^p\,dx,\,x,\,c\,x^n\right]$$

```
Int[Sec[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  x/(n*(c*x^n)^(1/n))*Subst[Int[x^(1/n-1)*Sec[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

```
Int[Csc[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  x/(n*(c*x^n)^(1/n))*Subst[Int[x^(1/n-1)*Csc[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

2. $\int (e\,x)^m \operatorname{Sec}[d\,(a+b\operatorname{Log}[c\,x^n])]^p\,dx$

1. $\int (e\,x)^m \operatorname{Sec}[d\,(a+b\operatorname{Log}[x])]^p\,dx$

1: $\int (e\,x)^m \operatorname{Sec}[d\,(a+b\operatorname{Log}[x])]^p\,dx$ when $p\in\mathbb{Z}$

– Derivation: Algebraic expansion

– Basis: $\operatorname{Sec}[d\,(a+b\operatorname{Log}[x])] == \frac{2\,e^{i\,a\,d}\,x^{i\,b\,d}}{1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}}$

– Basis: $\operatorname{Csc}[d\,(a+b\operatorname{Log}[x])] == -\frac{2\,i\,e^{i\,a\,d}\,x^{i\,b\,d}}{1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}}$

– Rule: If $p\in\mathbb{Z}$, then

$$\int (e\,x)^m \operatorname{Sec}[d\,(a+b\operatorname{Log}[x])]^p\,dx \longrightarrow 2^p\,e^{i\,a\,d\,p}\int \frac{(e\,x)^m\,x^{i\,b\,d\,p}}{\left(1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}\,dx$$

– Program code:

```
Int[(e_.*x_)^m_.*Sec[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  2^p*E^(I*a*d*p)*Int[(e*x)^m*x^(I*b*d*p)/(1+E^(2*I*a*d)*x^(2*I*b*d))^p,x] /;
FreeQ[{a,b,d,e,m},x] && IntegerQ[p]
```

```
Int[(e_.*x_)^m_.*Csc[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  (-2*I)^p*E^(I*a*d*p)*Int[(e*x)^m*x^(I*b*d*p)/(1-E^(2*I*a*d)*x^(2*I*b*d))^p,x] /;
FreeQ[{a,b,d,e,m},x] && IntegerQ[p]
```

2: $\int (e\,x)^m \operatorname{Sec}[d\,(a+b\operatorname{Log}[x])]^p\,dx$ when $p \notin \mathbb{Z}$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $\partial_x \frac{\operatorname{Sec}[d\,(a+b\operatorname{Log}[x])]^p\,(1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d})^p}{x^{i\,b\,d\,p}} == 0$

Basis: $\partial_x \frac{\operatorname{Csc}[d\,(a+b\operatorname{Log}[x])]^p\,(1-e^{2\,i\,a\,d}\,x^{2\,i\,b\,d})^p}{x^{i\,b\,d\,p}} == 0$

Rule: If $p \notin \mathbb{Z}$, then

$$\int (e\,x)^m \operatorname{Sec}[d\,(a+b\operatorname{Log}[x])]^p\,dx \longrightarrow \frac{\operatorname{Sec}[d\,(a+b\operatorname{Log}[x])]^p\,\left(1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}{x^{i\,b\,d\,p}} \int \frac{(e\,x)^m\,x^{i\,b\,d\,p}}{\left(1+e^{2\,i\,a\,d}\,x^{2\,i\,b\,d}\right)^p}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*Sec[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  Sec[d*(a+b*Log[x])]^p*(1+E^(2*I*a*d)*x^(2*I*b*d))^p/x^(I*b*d*p)*
    Int[(e*x)^m*x^(I*b*d*p)/(1+E^(2*I*a*d)*x^(2*I*b*d))^p,x] /;
FreeQ[{a,b,d,e,m,p},x] && Not[IntegerQ[p]]
```

```
Int[(e_.*x_)^m_.*Csc[d_.*(a_.+b_.*Log[x_])]^p_.,x_Symbol] :=
  Csc[d*(a+b*Log[x])]^p*(1-E^(2*I*a*d)*x^(2*I*b*d))^p/x^(I*b*d*p)*
    Int[(e*x)^m*x^(I*b*d*p)/(1-E^(2*I*a*d)*x^(2*I*b*d))^p,x] /;
FreeQ[{a,b,d,e,m,p},x] && Not[IntegerQ[p]]
```

2: $\int (e\,x)^m \operatorname{Sec}\left[d\,\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p\,dx$

- Derivation: Piecewise constant extraction and integration by substitution
- Basis: $\partial_x \frac{x}{\left(c\,x^n\right)^{1/n}} = 0$
- Basis: $\frac{F\left[c\,x^n\right]}{x} = \frac{1}{n}\operatorname{Subst}\left[\frac{F[x]}{x},\ x,\ c\,x^n\right]\partial_x\left(c\,x^n\right)$
- Rule:

$$\int (e\,x)^m \operatorname{Sec}\left[d\,\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p\,dx \longrightarrow \frac{(e\,x)^{m+1}}{e\,\left(c\,x^n\right)^{(m+1)/n}}\int \frac{\left(c\,x^n\right)^{(m+1)/n}\operatorname{Sec}\left[d\,\left(a+b\operatorname{Log}\left[c\,x^n\right]\right)\right]^p}{x}\,dx$$

$$\longrightarrow \frac{(e\,x)^{m+1}}{e\,n\,\left(c\,x^n\right)^{(m+1)/n}}\operatorname{Subst}\left[\int x^{(m+1)/n-1}\operatorname{Sec}\left[d\,\left(a+b\operatorname{Log}[x]\right)\right]^p\,dx,\ x,\ c\,x^n\right]$$

```
Int[(e_.*x_)^m_.*Sec[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  (e*x)^(m+1)/(e*n*(c*x^n)^((m+1)/n))*Subst[Int[x^((m+1)/n-1)*Sec[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

```
Int[(e_.*x_)^m_.*Csc[d_.*(a_.+b_.*Log[c_.*x_^n_.])]^p_.,x_Symbol] :=
  (e*x)^(m+1)/(e*n*(c*x^n)^((m+1)/n))*Subst[Int[x^((m+1)/n-1)*Csc[d*(a+b*Log[x])]^p,x],x,c*x^n] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && (NeQ[c,1] || NeQ[n,1])
```

Rules for integrands of the form u Trig[a x^n Log[b x]] Log[b x]

1. $\int$ u Sin[a x^n Log[b x]] Log[b x] dx

1: $\int$ Sin[a x Log[b x]] Log[b x] dx

Rule:

$$\int \text{Sin}[a\,x\,\text{Log}[b\,x]]\,\text{Log}[b\,x]\,dx \longrightarrow -\frac{\text{Cos}[a\,x\,\text{Log}[b\,x]]}{a} - \int \text{Sin}[a\,x\,\text{Log}[b\,x]]\,dx$$

Program code:

```
Int[Sin[a_.*x_*Log[b_.*x_]]*Log[b_.*x_],x_Symbol] :=
  -Cos[a*x*Log[b*x]]/a - Int[Sin[a*x*Log[b*x]],x] /;
FreeQ[{a,b},x]
```

```
Int[Cos[a_.*x_*Log[b_.*x_]]*Log[b_.*x_],x_Symbol] :=
  Sin[a*x*Log[b*x]]/a - Int[Cos[a*x*Log[b*x]],x] /;
FreeQ[{a,b},x]
```

2: $\int x^m \operatorname{Sin}[a x^n \operatorname{Log}[b x]] \operatorname{Log}[b x] \, dx$ when $m == n - 1$

Rule: If $m == n - 1$, then

$$\int x^m \operatorname{Sin}[a x^n \operatorname{Log}[b x]] \operatorname{Log}[b x] \, dx \longrightarrow -\frac{\operatorname{Cos}[a x^n \operatorname{Log}[b x]]}{a n} - \frac{1}{n} \int x^m \operatorname{Sin}[a x^n \operatorname{Log}[b x]] \, dx$$

Program code:

```
Int[x_^m_.*Sin[a_.*x_^n_.*Log[b_.*x_]]*Log[b_.*x_],x_Symbol] :=
  -Cos[a*x^n*Log[b*x]]/(a*n) - 1/n*Int[x^m*Sin[a*x^n*Log[b*x]],x] /;
FreeQ[{a,b,m,n},x] && EqQ[m,n-1]
```

```
Int[x_^m_.*Cos[a_.*x_^n_.*Log[b_.*x_]]*Log[b_.*x_],x_Symbol] :=
  Sin[a*x^n*Log[b*x]]/(a*n) - 1/n*Int[x^m*Cos[a*x^n*Log[b*x]],x] /;
FreeQ[{a,b,m,n},x] && EqQ[m,n-1]
```